

Exercises with AMPL

Antonio Mucherino

Laboratoire d'Informatique, École Polytechnique

Course on Operations Research (ISC610A)

Semester I - 2008/09

TD3 - November 20th 2008

Buying metro tickets

One day, before going out, I thought:

*“I have to take the metro, and I have no tickets, but I have several coins that I can use for buying a **carnet** of tickets”.*

Then, I went to the metro station with all my coins. I selected the *carnet* of tickets I wanted from the machine, and I started to put my coins inside, starting from the smallest ones, in order to get rid of as many coins as I could:

“This is a very good way to recycle my coins!”

I was thinking while proudly I was grabbing coins from my bag.
But **I was about to discover something very bad!**

Buying metro tickets

*You cannot use more than 20 coins per transaction.
Transaction cancelled.*

WHAT? I thought. My enthusiasm faded out!

I had inserted no more than 5 euros so far, because I was choosing all the smallest coins, while I had to pay **11.40 euros** in total!

The machine started to give me back all the coins, and, during that tragic moment, my mind was looking for a solution.

- **Solution 1:** I may mainly use 1 or 2 euro coins.
- **But**, in this way, I would not use many coins.
- **Solution 2:** I may ask my friend **AMPL** to help me.

“I need to run back home and write a precise request!”

The request

What is my problem exactly?

- I want to get rid of as many coins as possible.
- The sum of the values of all the used coins must be the *carnet* price (11.40 euros).
- The total number of coins cannot be more than 20.
- I can only use the coins that I actually have.
- I need to tell my friend how many coins I have for each given value.

The mathematical model

Parameters

- N , integer, the number of coin values
- $totValue$, real, the price to be paid
- C , vector of length N , the values of the different coins
- nC , vector of length N , the number of coins available for a given value

Variables

- x_j , integer, indicates how many coins are used for a given value

The mathematical model

Objective function

- “I want to get rid of as many coins as possible”

$$\max \sum_{i=1}^N x_i$$

Constraints

- “The total sum must be 11.40” $\sum_{i=1}^N C_i \times x_i = \text{totValue}$

- “The total number of coins is limited to 20” $\sum_{i=1}^N x_i \leq 20$

- “I can't use coins I don't have” $\forall i \in \{1, 2, \dots, n\} \quad x_i \leq nC_i$

carnet.mod

The `.mod` file contains the **model** of the problem to be solved:

```
# carnet.mod

param N integer;
param totValue;
param C {1..N};
param nC {1..N};
var x {1..N} integer, >= 0;

maximize numCoins : sum{i in 1..N} x[i];

subject to carnet_price : sum{i in 1..N} C[i] * x[i] = totValue;

subject to overall_limit : sum{i in 1..N} x[i] <= 20;

subject to limit {i in 1..N} : x[i] <= nC[i];
```

carnet.dat

The `.dat` file contains the **data** of the problem to be solved:

```
# carnet.dat  
  
param N := 6;  
param totValue := 11.40;  
  
param : C nC :=  
1 0.05 6  
2 0.10 8  
3 0.20 4  
4 0.50 4  
5 1.00 7  
6 2.00 5  
;
```


carnet.run

The `.run` file refers to the other two text files
(the `model` and the `data`),
sets the solver to be used,
and displays the final result.

```
# carnet.run  
  
model carnet.mod;  
data carnet.dat;  
option solver cplex;  
  
solve;  
  
display numCoins;  
display sum{i in 1..N} C[i] * x[i];  
display x;
```

Calling AMPL

```
[antonio.mucherino@ferrari ~/ampl]$ cat carnet.run | ampl
ILOG AMPL 10.100, licensed to "ecolepolytechnique-palaiseau".
AMPL Version 20060626 (Linux 2.6.9-5.ELsmp)
ILOG CPLEX 10.100, licensed to "ecolepolytechnique-palaiseau", options: e m b q use=8
CPLEX 10.1.0: optimal integer solution; objective 20
4 MIP simplex iterations
0 branch-and-bound nodes
numCoins = 20

sum{i in 1 .. N} C[i]*x[i] = 11.4

x [*] :=
1 6
2 3
3 4
4 0
5 4
6 3
;

[antonio.mucherino@ferrari ~/ampl]$
```

The proposed exercises can be downloaded from:

www.antoniomucherino.it